Extrinsic functions in GAMS

Toni Lastusilta[1], Michael Bussieck[*], Stefan Emet[2]

[1,*] GAMS Software GmbH, TLastusilta@gams.com, MBussieck@gams.com

[2] Department of mathematics and statistics, University of Turku, Finland, SEmet@utu.fi

Hannover, Germany September 2012

Agenda

- Functions and Equations in GAMS
- Chromatographic Separation
- Curve fitting
- Conclusion

A simple GAMS model




```
$Title Nonlinear Test Problem - Multiple Solutions (MHW4DXX,SEQ=295)
set i /1*5/
Variables m, x(i);
Equations funct, eq1, eq2, eq3;

funct.. m =e= sqr(x('1')-1) + sqr(x('1')-x('2'))+ power(x('2')-x('3'),3)
            + power(x('3')-x('4'),4) + power(x('4')-x('5'),4) ;

eq1.. x('1') + sqr(x('2')) + power(x('3'),3) =e= 3*sqrt(2) + 2 ;

eq2.. x('2') - sqr(x('3')) + x('4')          =e= 2*sqrt(2) - 2 ;

eq3.. x('1')*x('5') =e= 2 ;

Model wright / all / ;

x.l('1') = -1; x.l('2') = 2; x.l('3') = 1; x.l('4') = -2; x.l('5') = -2;

Solve wright using nlp minimizing m;
```

A simple GAMS model



Intrinsic function

- For (MI)NLP Models GAMS provides underlying solvers with point evaluations for:
  - Function value
  - Gradient
  - Hessian

Classification of functions

Type	Function	Derivative	Examples
Smooth	Continuous	Continuous	exp, sin, log
Non-Smooth	Continuous	Discontinuous	max, min, abs
Discontinuous	Discontinuous	Discontinuous	ceil, sign

Extrinsic functions and external equations

- GAMS users can write their own extrinsic functions and external equations.
- External equations was introduced in GAMS in 1996
- Extrinsic functions and the GAMS Function Library Facility was introduced in GAMS 23.7 in 2011
- Function Libraries
  - Fitpack
  - Piecewise Polynomial Library
  - Stochastic Library
  - LINDO Sampling Library

Extrinsic function vs. external equation

- Example 1

$$f(x_1, x_2) = z, \qquad f(x_1, x_2) = \int_{x_1}^{x_2} x \, dx = \frac{1}{2}(x_2^2 - x_1^2)$$

- as extrinsic function:

```
e1.. integralx(x1,x2) =E= z;
```

- as external equation:

```
e1.. 1*x1 + 2*x2 + 3*z =X= 1;
```

Extrinsic function vs. external equation

- Example 2 $$\int_{x_1}^{(x_2)^2} g(x, y_i)dx \geq (x_2 - x_1)^2 \quad \forall\, i, i = 1,2,3,4$$

  - as extrinsic function:

```
e(i).. integral(x1,sqr(x2),y(i)) =g= sqr(x2-x1);
```

  - as external equation:

```
e(i).. 1*x1 + 2*x2 + (2+ord(i))*y(i)
          + (2+card(i)+ord(i))*slack(i) =X= ord(i);
                          or
e1.. 1*x1 + 2*x2 + 3*y('1') + 7 *slack('1') =X= 1;
e2.. 1*x1 + 2*x2 + 4*y('2') + 8 *slack('2') =X= 2;
e3.. 1*x1 + 2*x2 + 5*y('3') + 9 *slack('3') =X= 3;
e4.. 1*x1 + 2*x2 + 6*y('4') + 10*slack('4') =X= 4;
```

A characteristic comparison

Characteristic	Extrinsic function	External equation
Argument limitation	10	No
Available in statements	Yes	No
Debugging support	Yes	No
Returns Hessian to solver	Yes	No

- Argument limitation: `integral(x1,x2,x3x,x4,x5,x6,x7,x8,x9,x10)`
- Available in statements: `x.l=sqr(integralx(0,2));`
- Debugging support: support for numerical gradient and hessian

  `integralx.grad(2:0,2)  vs.  integralx.gradn(2:0,2)`
- CONOPT verifies numerically derivative information
- Many solvers can solve without Hessian information
- Global solvers (BARON, Couenne , LindoGlobal, SCIP) analyze the algebra and can't therefore handle extrinsic functions or external equations.

GAMS

Agenda

- Functions and Equations in GAMS
- **Chromatographic Separation**
- Curve fitting
- Conclusion

Chromatographic Separation

- Separation of a multicomponent mixture
  - Example, glucose flows faster through a column than fructose

- Separation can be done by Simulated Moving Bed Chromatography

- To model the process a boundary value problem needs to be solved
  - A boundary value problem: Differential Equation (DE) together with a set of additional constraints

A one column system





$$\frac{\partial c_{kj}}{\partial t} + F\,\frac{\partial q_{kj}}{\partial t} + u\,\frac{\partial c_{kj}}{\partial z} = D_j\,\frac{\partial^2 c_{kj}}{\partial z^2}$$



Goal: Maximize the profits during a cycle, i.e.

max 1/T*(incomes-costs)

$$\max\ \frac{1}{\tau}\left(\sum_{i=1}^{T} y \int_{t_{i-1}}^{t_i} c(t, z_H)\,dt - y\,\left(t_i - t_{i-1}\right)\right)$$

Simulated Moving Bed (SMB)

<u>Purity requirements:</u>

90% of product 1

90% of product 2.

Collect separated products



Chromatographic Separation

- The mass (m) is calculated by integration over time
  - Pure: $m_{ij} \le \int_{t_{i-1}}^{t_i} c_j(t, z_H)\,dt + M(1 - y_{ij})$
  - Impure: $m_{ij} \ge \int_{t_{i-1}}^{t_i} c_j(t, z_H)\,dt - M(1 - \sum_{\substack{l=1 \\ l \ne j}}^{C} y_{il})$
  - c=concentration
  - t=time
  - Z_H=column height
- The concentration is obtained by solving a boundary value problem including a DE (Guiochon et al., 1994):

$$\left(1 + F\beta_j + F\sum_{l=1}^{C} \beta_{jl} c_l\right)\frac{\partial c_j}{\partial t} + F c_j \sum_{l=1}^{C} \beta_{jl} \frac{\partial c_l}{\partial t} + u\frac{\partial c_j}{\partial z} = D_j \frac{\partial^2 c_j}{\partial z^2}$$

CS using extrinsic function

- Pure: $$m_{ij} \le \int_{t_{i-1}}^{t_i} c_j(t, z_H)dt + M(1 - y_{ij})$$
- Impure: $$m_{ij} \ge \int_{t_{i-1}}^{t_i} c_j(t, z_H)dt - M(1 - \sum_{\substack{l=1 \\ l \ne j}}^{C} y_{il})$$

- c=concentration
- t=time
- Z_H=column height

```
e1(i,j).. m(i,j) =L=
PDE(ord(i),ord(j),t('1'),t('2'),t('3'),t('4'),yf('2'),yf('3'),yf('
4')) + BigM*(1-y(i,j));


e2(i,j).. m(i,j) =G=
PDE(ord(i),ord(j),t('1'),t('2'),t('3'),t('4'),yf('2'),yf('3'),yf('
4')) - BigM*(1-sum(jP$(not sameas(j,jP)),y(i,jP)) );
```

- Note that the feed choise yf is required as input to solve the boundary value problem.

CS using external equations

```
startt        = 0;
startyf       = startt      + card(i)-1;
startyu       = startyf     + card(i);
startm        = startyu     + card(i)*card(j);
starte1slack  = startm      + card(i)*card(j);
starte2slack  = starte1slack + card(i)*card(j);
e1(ijii(i,j,ii))..
       sum(iP, (startt+ord(iP))*t(iP))
      +sum(iP$(ord(iP)>1), (startyf+ord(iP))*yf(iP))
      +(startyu+ord(ii))*y(i,j)
      +(startm+ord(ii))*m(i,j)
      +(starte1slack+ord(ii))*e1slack(ii) =x= ord(ii);
e2(ijii(i,j,ii))..
       sum(iP, (startt+ord(iP))*t(iP))
      +sum(iP$(ord(iP)>1), (startyf+ord(iP))*yf(iP))
      +sum(ijiiP(i,jP,iiP)$(not sameas(j,jP)),(startyu+ord(iiP))*y(i,jP)
      +(startm+ord(ii))*m(i,j)
      +(starte2slack+ord(ii))*e2slack(ii)) =x= ord(ii)+card(ii);
```

CS using external equations

```
e1(i1,j1,ii1)..  (1)*t(i1) + (2)*t(i2) + (3)*t(i3) + (4)*t(i4) + (8)*y(i1,j1)
      + (16)*m(i1,j1) + (5)*yf(i2) + (6)*yf(i3) + (7)*yf(i4) + (24)*e1slack(ii1)
      =X= 1 ;

e1(i1,j2,ii2)..  (1)*t(i1) + (2)*t(i2) + (3)*t(i3) + (4)*t(i4) + (9)*y(i1,j2)

      + (17)*m(i1,j2) + (5)*yf(i2) + (6)*yf(i3) + (7)*yf(i4) + (25)*e1slack(ii2)
      =X= 2 ;
e1(i2,j1,ii3)..  (1)*t(i1) + (2)*t(i2) + (3)*t(i3) + (4)*t(i4) + (10)*y(i2,j1)

      + (18)*m(i2,j1) + (5)*yf(i2) + (6)*yf(i3) + (7)*yf(i4) + (26)*e1slack(ii3)
      =X= 3 ;
e1(i2,j2,ii4)..  (1)*t(i1) + (2)*t(i2) + (3)*t(i3) + (4)*t(i4) + (11)*y(i2,j2)

      + (19)*m(i2,j2) + (5)*yf(i2) + (6)*yf(i3) + (7)*yf(i4) + (27)*e1slack(ii4)
      =X= 4 ;
```

Observations

- The extrinsic functions provides a more intuitive understanding of the algebra and are better supported
- For the studied chromatographic separation problem the extrinsic function input limitation limits the use to a one column system
- For a general model including multiple columns the external equation formulation is more appropriate

Agenda

- Functions and Equations in GAMS
- Chromatographic Separation
- **Curve fitting**
- Conclusion

Curve fitting

- Usage:
  - Only a number of data points is known (sampling or experimentation)
  - Approximation of a computationally demanding function
- A curve can be calculated by
  - Least square method
  - Polynomial interpolation
  - Spline interpolation
- Curve fitting with Fitpack: control of smoothness and fit
  - Implementation by splines and tensor product splines
  - automatic knot selection, error smoothing, data reduction
  - author: Paul Dierckx

Illustration 1



Illustration 2



Extrinsic function implementation

- Curve fitting with a extrinsic function

```
set n discretization points per variable /0*100/
    f function index                     / 1 /
    p datapoints                         / 1*7 /
    v vector[x y]                        / x,y /;
parameter fitdata(f,p,v) /1.1.x -3, 1.2.x -2, …/
parameter x1(n) fractional values within interval ;
parameter y1(n) spline function values for y ;
scalar rc;
$gdxout fit
$unload f p v fitdata
$gdxout

$funclibin fitlib fitfclib
function fitp /fitlib.fitparam/
         fit  /fitlib.fitfunc /;

x1(n) = -3 + 6 *(ord(n)-1)/(card(n)-1);
rc=fitp(1,1,100);
y1(n)=fit(1,x1(n));
```

External equation implementation

- Curve fitting with a external equation (E. Kalvelagen, 2002)

```
set d 'data points' /d1*d7/;
parameter datax(d) 'data points, x coordinate' / d1 -3, …/
parameter datay(d) 'data points, y coordinate' / d1 0 , …/
variables z,x,y;

file datafile /'splinedata.txt'/; put datafile;
put ns:10:0,nf:10:0/;
…
equations zdef,spline;
zdef..   z =e= y;
spline.. 1 =x= 1*x+2*y;
…
set n discretization points per variable /0*100/
x1(n) = -3 + 6 *(ord(n)-1)/(card(n)-1);
loop(n,
  x.fx = x1(n) ;
  solve m minimizing z using nlp;
);
```

Agenda

- Functions and Equations in GAMS
- Chromatographic Separation
- Curve fitting
- Conclusion

Conclusions

- **Extrinsic functions** are more intuitive to use and are better supported
- For the studied chromatographic separation problem the external equation formulation was more suitable
- Extrinsic function libraries can provide domain specific functionality, for example, curve fitting

More information on

- extrinsic functions: www.gams.com/dd/docs/bigdocs/GAMSUsersGuide.pdf
- external equations: www.gams.com/docs/externalequ.htm

Thank You !

USA

GAMS Development Corp.
1217 Potomac Street, NW
Washington, DC 20007
USA

Phone: +1 202 342 0180
Fax: +1 202 342 0181

http://www.gams.com
sales@gams.com
support@gams.com

Europe

GAMS Software GmbH
Eupener Str. 135-137
50933 Cologne
Germany

Phone: +49 221 949 9170
Fax: +49 221 949 9171

http://www.gams.com
info@gams.de
support@gams.com